
/**************************************************************************
*   This program will calculate the recodes for the public use file.1999 NATIONAL FINAL
*   1/8/01
*   Program:      pufrcd.sas
*   Programmed by: Diane Schwartz
*   Date:     January 20,1998
*   Updated:  April 9,1998
*   Revised:  May 19,1998 - corrected universe problem for zsmhc
*   Revised:  November 20, 1998 - included a new recode for HISTRY
*   Revised:  January 27, 1999 - corrected nummor according to new specs
*   Revised:  April 8, 1999 - corrected problem with nummor, not counting all 
*                             the cases that have no mortgage.
*   Revised:  April 20, 1999 - added new variable METRO2, because of a problem 
*                             with no geography conversion for 90 - 80 on 
*                             UACODE.  This new variable is a recode of METRO.
*   Revised:  May 24, 1999 - added new variable METRO3 and took out the 
*                            variable METRO2
*
*   Input Library:  /natdata/(survey year)/topcodes
*                   /natdata/(survey year)/bridge
*                   /natdata/(survey year)/master.ssd01
*                   /natdata/(survey year)/houshldr
*
*   Output Library: /natadata/(survey year)/recodes
*
*   The following recodes are coded in this program:
*                  cmsa  - renamed from cmsa80
*                  smsa  - renamed from pmsa80
*                  per
*                  zadult
*                  zinc
*                  zinc2
*                  phone 
*                  kitchen
*                  plumb
*                  rooms
*                  confee
*                  zadeq
*                  zmshc
*                  amtx
*                  nummor
*                  ran
*                  rac
*                  degree
*                  metro
*                  histry

*                  metro2  --  as of May 24, 1999 this recode is no longer done
*                  metro3   
*
****************************************************************************/

/* ** set up lib names ** */
libname bridg "$BRIDGE";
libname rcd   "$RECD";
libname mast "$MASTR";
libname topc "$TOPCD";

missing D R B X M N U _ ;

/*  sort datasets on control number 13 to enable views to be created */
/*
proc sort data=topc.houshld out=work.houshld;
  by ctrlnm13;
run;

proc sort data=topc.person out=work.person;
  by ctrlnm13 person;
run;

proc sort data=mast.master out=work.master;
  by ctrlnm13;
run;

proc sort data=topc.homimp out=work.homimp;
  by ctrlnm13;
run;

proc sort data=topc.mortg out=work.mortg;
  by ctrlnm13;
run;

data tmphhr(keep=ctrlnm13 phonenum);
   set bridg.houshldr;
run;
proc sort data=tmphhr out=houshldr;
  by crtlnm13;
run;

proc sort data=bridg.homimpr out=work.homimpr;
  by ctrlnm13;
run;
*/

/* ** create a view to process the recodes ** */
data viewed1 / view=viewed1;
   merge topc.houshld (in = hhldend
                       keep = ctrlnm13 amte amtf amtg amti amto amtt amtw
                              amtxq amtxre badstep bedrms bigp burner 
                              camf climb confeeq cook cracks dens dining elev
                              famrm flrent fmhotf freeze frent hequip prent
                              holes hotpip ileak kexclu kitch leak living
                              lrent ltsok m3rod mhotfe mice nowire numblow
                              numcold numtlt nunit2 othfn oven plugs
                              railok rats recrm refr rent sharpf sink
                              status tenure toilet tub txre vacancy
                              vother curper buye cstmnt baths hel 
                              henum mcnt mg telhh telav)
     topc.person (in = endper
                  keep = ctrlnm13 age hhmem rel sal);
        by ctrlnm13;
     if not (endper and hhldend) then
     do;
       if endper then
       do;
/*   put 'no houshld obs for persons ctrlnm13 = ' ctrlnm13;  */
        cnohh+1;
       end;
       else
       do;
/*   put 'no persons obs for houshld ctrlnm13 = ' ctrlnm13;  */
       cnopr+1;
      end;
    end;
run;

data viewed2 / view=viewed2;
   merge mast.master (in=endmst
                        keep = ctrlnm13 cmsa80 region pmsa80 coolday
                               heatday msasta80 uacode80 urbrur80
                               convdate
                               rename = (cmsa80=cmsa pmsa80=smsa))
         viewed1 (in=endvw);
   by ctrlnm13;
   if not (endmst and endvw) then
   do;
    if endvw then put 'no master obs for viewed ctrlnm13 = ' ctrlnm13;
    else delete;

   end;
run;

data viewed3 / view = viewed3;
    merge topc.homimp (in=hiend
                     keep = ctrlnm13 adbedam adbtham addbed addbth
                            addkit addoth adkitam adotham garag prch
                            deck carport othout bthroom kitchn siding 
                            bthrdam doorwin insulat
                            wtrpipe plumfxt wiring carpet floorin
                            panel cenair heatequ septic wtrhtr dshwshr
                            dsposal kitrdam rnbtham sidam dooram inslam
                            pipeam plfxtam wiram subflam flooram panelam
                            cenaram htequam crprtam deckam disrpam 
                            driveam drivewy dshwam dsplam fence fenceam
                            finflam garagam insaam insbam inscam 
                            nwbed nwbedam nwbth nwbtham nwoth nwotham
                            nwrec nwrecam othotam outaam outbam outcam
                            patio patioam pool poolam porcham 
                            renbedr renoth rnbedam rnotham roof roofam
                            septam shed shedam wtrham quake tornado lndsld
                            fire flood othdis secure securam)
          viewed2 (in=endvw2);
    by ctrlnm13;
    if not (hiend and endvw2)then
    do;
      if hiend then put 'no viewed2 obs for homimp ctrlnm13 = ' ctrlnm13;
    end;
run;

data viewed4 / view = viewed4;
    merge topc.homimp (in=hirend keep = ctrlnm13 misina misinb misinc misouta 
                         misoutb misoutc) 
          viewed3 (in=endvw3);
    by ctrlnm13;
    if not (hirend and endvw3) then
    do;
      if hirend then put 'no viewed3 obs for homimpr ctrlnm13 = ' ctrlnm13;
    end;
run;

data viewed5 /view = viewed5;
    merge bridg.houshldr(in=hhrend keep=ctrlnm13 phonenum)
          viewed4 (in=endvw4);
    by ctrlnm13;
    if not (hhrend and endvw4) then

    do;
      if hhrend then put 'no viewed4 obs for hhoushldr ctrlnm13 = ' ctrlnm13;
    end;
run;

data viewedt / view = viewedt;
    merge topc.mortg (in=mtgend
                    keep = ctrlnm13 amtm amtm2 hepmt1 hepmt2 hepmt3
                           inspmt inpmt2 pmt pmt2 pmt3 pmt4 
                           taxpmt txpmt2)
          viewed5 (in=endvwd);
    by ctrlnm13;
    if not (mtgend and endvwd) then
    do;
      if mtgend then put 'no viewedt obs for mortg ctrlnm13 = ' ctrlnm13;
    end;
run;

/* ** this section creates a data set with the recodes included. ** */
data rcd.pufrcd (keep = ctrlnm13 per zadult zinc zinc2 phone kitchen plumb
                          rooms confee zadeq zsmhc amtx nummor ran rac 
                          region smsa cmsa degree metro histry metro3);

    length ctrlnm13 $13 per 3. zadult 3. zinc 5. zinc2 5. phone $1 kitchen $1
           plumb $1 rooms 3. confee 3. zadeq $1 zsmhc 4. amtx 4. nummor 4.
           ran 4. rac 5. region $1 smsa 4. cmsa $2 degree $1 metro $1
           histry $1 srvyr $4 metro3 $1;

    set  viewedt  end=send;
      by ctrlnm13;

    retain incnt 0;
    retain ctrlno 0;
    retain ctllst 0;
    retain outcnt 0;
    incnt+1;

/*    if(mod(incnt,500) = 0) then put 'processing obs number ' incnt; */
    if first.ctrlnm13 then ctrlno+1;
    if last.ctrlnm13 then ctllst+1;

 if first.ctrlnm13 then 
 do;
   per = .B;

   zadult = .B;
   zinc = .B;
   zinc2 = .B;
   rooms = .B;
   confee = .B;
   zsmhc = .B;
   amtx = .B;
   ran = .B;
   rac = .B;
   plumb = 'B';
   zadeq = 'B';
   phone = 'B';
   kitchen = 'B';
   metro = ' ';
   metro3 = ' ';
 end;

 if status = '1' then
 do;

    if first.ctrlnm13 then
    do;
      per = 0;
      zadult = 0;
      zinc = 0;
      zinc2 = 0;
      i = 1;
      if (-999998 <= vother <= 999998) then
      do;
         zinc+vother;
         zinc2+vother;
      end;
    end;

/* per and zadult recodes  - number of persons and number of adults over 18 */

       if hhmem='1' then
       do;

         per+1;
         if (age >= 18) then
         do;
           zadult+1;
/*           if ctllst < 10 then
            put 'zadult = ' zadult ' and age = ' age; */

         end;

/* zinc recode - income of reference person and household
   members related to reference person  */

       if (age >= 16 and (rel = 1 or rel = 2 or rel = 20 or (rel >= 22 and
          rel <= 26))) then
          zinc+sal;

/* zinc2 recode - household income */

       if (age >= 16) then
          zinc2+sal;

       end;  /* end of hhmem check */

 end;   /* end of status check */

 if status > '0' and status < '4' then
 do;

/* phone recode - */
   if (status = '1') then do;
      if (phonenum ne ' ' or telhh = 1 or telav = 1) then phone = '1';
      else if (telhh = 2 and telav = 2) then phone = '2';
      else phone = 'D';
   end;

/* kitchen recode  - complete kitchen facilities */
    kitchen = ' ';

    if(((nunit2 = '1' or nunit2 = '2' or nunit2 = '4')
         and (sink = '1')
         and (refr = '1')
         and (cook = '1' or burner = '1' or oven = '1'))
       or
         ((nunit2 = '3' or nunit2 = '5')
         and (kexclu = '1')
         and (sink = '1')
         and (refr = '1')
         and (cook = '1' or burner = '1' or oven = '1'))) then
     do;
        kitchen = '1';
     end;
     else

     do;
        if ((sink = '2')
           or (refr = '2')
           or (cook = '2' and burner = '2' and oven = '2')
           or ((nunit2 = '3' or nunit2 = '5') and (kexclu = '2'))) then
           do;
              kitchen = '2';
           end;
     end;

/* plumb recode  - exclusive use of plumbing */
   plumb = ' ';

   if ( hotpip = '1' and toilet = '1' and tub = '1' and sharpf = '2') then
   do;
      plumb = '1';
   end;
   if ( hotpip = '2' or toilet = '2' or tub = '2' or sharpf = '1')then
   do;
     plumb = '2';
   end;

/* rooms recode - number of rooms */

   rooms = 0;
   rooms = sum(bedrms,kitch,living,dining,famrm,recrm,dens,othfn);
   
/*   if (status eq 1 and rooms eq 0) then rooms = 1;  */
   if rooms >= 21 then rooms = 21;

/* confee recode - condominium or homeowners assoc. or mobile home monthly
                    fees   */

   if confeeq = . then confee = .;
   if confeeq = 0 then confee = 0;
   if confeeq >= 1 then
   do;
     confee = (confeeq * camf) / 12;
     confee = confee - 5;
     confee = round(confee,10);
     confee+5;
  end;

end;      /* end of status check */

/* zadeq recode  - adequacy of housing */
if status = '1' then
do;
  zadeq = ' ';

  if (baths < 2 and (hotpip = '2'
   or tub = '2'
   or toilet = '2'
   or sharpf = '1'))
   or
     (freeze = '1'
   and (numcold >= '3' and numcold <= '8'))
   or
     (buye = '1')
   or
     (nowire = '2'
   and plugs = '2'
   and (numblow >= '3' and numblow <= '8')) then
   do;
      zadeq = '3';
   end;

   n = 0;
   if leak = '1' then n+1;
   if ileak = '1' then n+1;
   if holes = '1' then n+1;
   if cracks = '1' then n+1;
   if bigp = '1' then n+1;
   if rats = 'X' then n+1;
   if n >= 5 then
      zadeq = '3';

   m = 0;
   if ltsok = '4' or ltsok = '5' then m+1;
   if badstep = '2' then m+1;
   if railok = '1' or railok = '3' then m+1;
   if climb >= 3 and elev ne '2' then m+1;
   if m = 4 then
      zadeq = '3';

   if zadeq ne '3' then
   do;

     if numtlt >= '3' and numtlt <= '8' then
         zadeq = '2';
     if hequip = 7  then
         zadeq = '2';

     n=0;
     if leak = '1' then n+1;
     if ileak = '1' then n+1;
     if holes = '1' then n+1;
     if cracks = '1' then n+1;
     if bigp = '1' then n+1;
     if rats = 'X' then n+1;
     if n = 3 or n = 4 then
          zadeq = '2';

     m=0;
     if ltsok = '4' or ltsok = '5' then m+1;
     if badstep = '2' then m+1;
     if railok = '1' or railok = '3' then m+1;
     if climb >= 3 and elev ne '2' then m+1;
     if m = 3 then
          zadeq = '2';

     if kitchen = '2' then
          zadeq = '2';
  end;

     if zadeq ne '3' and zadeq ne '2' then
          zadeq = '1';
end;  /* end of status check for zadeq universe */

/* zsmhc - monthly housing costs recode  */

if amtxre < 0 then amtxre = 0;
if mhotfe < 0 then mhotfe = 0;
if cstmnt < 0 then cstmnt = 0;

if status = '1' then
do;

  zsmhc = 0;

  if (1 <= amte <= 998) then zsmhc+amte;
  if (1 <= amtg <= 998) then zsmhc+amtg;
  if (amto >= 4) then zsmhc+(amto / 12);

  if (amtf >= 4) then zsmhc+(amtf / 12);
  if (amtt >= 4) then zsmhc+(amtt / 12);
  if (amtw >= 4) then zsmhc+(amtw / 12);

  if ((amtxq >= 0) and (taxpmt ne '1' and txpmt2 ne '1')) then
  do;
    zsmhc+((amtxq - amtxre) / 12);
  end;

  if (amti >= 1) and (inspmt ne '1' and inpmt2 ne '1') then
  do;
    zsmhc+(amti / 12);
  end;

  if (confeeq >= 1) then
     zsmhc+((confeeq*camf)/12);
  if ( 1 <= lrent <= 1996) then
     zsmhc+((lrent * flrent) / 12);
  if (nunit2 = '4' or nunit2 = '5') then
  do;
    if (mhotfe >= 1) then
       zsmhc+((mhotfe * fmhotf) / 12);
  end;
  if ((rent >= 2) and (prent < 1 or prent > 9997)) then
     zsmhc+((rent*frent)/12);
  if (prent >= 1 and frent >= 1) then
   do;
     zsmhc+((prent*frent) / 12);
   end;
  if (pmt >= 1) then
     zsmhc+pmt;
  if (pmt2 >= 1) then
     zsmhc+pmt2;
  if (pmt3 >= 1) then
     zsmhc+pmt3;
  if (pmt4 >= 1) then
     zsmhc+pmt4;
/*  if (hepmt1 >= 1) then
     zsmhc+hepmt1; */
/*  if (hepmt2 >= 1) then
     zsmhc+hepmt2; */
/*  if (hepmt3 >= 1) then
     zsmhc+hepmt3; */
  if (amtm >= 1) then
     zsmhc = zsmhc-(amtm / 12);
  if (amtm2 >= 1) then

     zsmhc = zsmhc-(amtm2 / 12);
  if tenure = '1' then 
  do;
    if ((cstmnt / 12) >= 1) then
       zsmhc+(cstmnt / 12);
  end;

  zsmhc = round(zsmhc,1);
end;  /* end of universe check for zsmhc */

/* amtx recode - yearly real estate taxes */

amtx = .;
if((status = '1' and tenure = '1') or ((status = '2' or status = '3') and
  (vacancy = 3 or vacancy = 5))) then
  do;
    if ((status = '1' and tenure = '1') and txre = '2' and amtxq = 0) then
       amtx = 0;
    else
       if((status = '1' and tenure = '1') and txre = '1' and
         ((amtxq - amtxre) <= 0)) then
           amtx = 0;
    else
       if((status = '2' or status = '3') and (vacancy = 3 or vacancy = 5)
         and amtxq = 0) then
           amtx = 0;
    else
       if((status = '1' and tenure = '1') and txre = '2' and amtxq >= 1)
         then
           amtx = amtxq;
    else
       if((status = '1' and tenure = '1') and txre = '1' and
         ((amtxq - amtxre) >= 1)) then
           amtx = amtxq - amtxre;
    else
       if((status = '2' or status = '3') and (vacancy = 3 or
          vacancy = 5) and amtxq >= 1) then
           amtx = amtxq;

    if amtx >= 1 then
    do;
       amtx = amtx - 50;
       amtx = round(amtx,100);
       amtx+50;
    end;

end;  /* end of universe check for amtx */

/* ran recode - repair and alteration number  */

if status = '1' and tenure = '1' then
do;
  ran = 0;
  if (quake = 'X' or tornado = 'X' or lndsld = 'X' or
      fire = 'X' or flood = 'X' or othdis = 'X') then
           ran+1;

  if nwbth = 'X' then ran+1;
  if nwbed = 'X' then ran+1;
  if nwrec = 'X' then ran+1;
  if nwoth = 'X' then ran+1;

  if addbth = 'X' then ran+1;
  if addkit = 'X' then ran+1;
  if addbed = 'X' then ran+1;
  if addoth = 'X' then ran+1;

  if garag = 'X' then ran+1;
  if prch = 'X' then ran+1;
  if deck = 'X' then ran+1;
  if carport = 'X' then ran+1;
  if othout = 'X' then ran+1;
  if bthroom = 'X' then ran+1;
  if kitchn = 'X' then ran+1;

  if renbedr = 'X' then ran+1;
  if renoth = 'X' then ran+1;

  if roof = '1' then ran+1;
  if siding = '1' then ran+1;  
  if doorwin = '1' then ran+1;
  if insulat = '1' then ran+1;
  if wtrpipe = '1' then ran+1;
  if plumfxt = '1' then ran+1;
  if wiring = '1' then ran+1;

  if secure = '1' then ran+1;
  if carpet = '1' then ran+1;
  if floorin = '1' then ran+1;
  if panel = '1' then ran+1;
  if cenair = '1' then ran+1;

  if heatequ = '1' then ran+1;
  if septic = '1' then ran+1;
  if wtrhtr = '1' then ran+1;
  if dshwshr = '1' then ran+1;
  if dsposal = '1' then ran+1;
  
  if misina ne ' ' then ran+1;
  if misinb ne ' ' then ran+1;
  if misinc ne ' ' then ran+1;

  if drivewy = '1' then ran+1;
  if fence = '1' then ran+1;
  if patio = '1' then ran+1;
  if pool = '1' then ran+1;
  if shed = '1' then ran+1;
  
  if misouta ne ' ' then ran+1;
  if misoutb ne ' ' then ran+1;
  if misoutc ne ' ' then ran+1;
  

end;  /* end of universe check for ran */

/* rac recode - repair and alteration costs */

if status = '1' and tenure = '1' then
do;
   rac = 0;

   if kitrdam >= 2 then rac+kitrdam; 
   if bthrdam >= 2 then rac+bthrdam;
   if disrpam >= 2 then rac+disrpam;
   if nwbtham >= 2 then rac+nwbtham;
   if nwbedam >= 2 then rac+nwbedam;
   if nwrecam >= 2 then rac+nwrecam;
   if nwotham >= 2 then rac+nwotham;

   if adbtham >= 2 then rac+adbtham;
   if adkitam >= 2 then rac+adkitam;
   if adbedam >= 2 then rac+adbedam;
   if adotham >= 2 then rac+adotham;

   if garagam >= 2 then rac+garagam;
   if porcham >= 2 then rac+porcham;
   if deckam >= 2 then rac+deckam;
   if crprtam >= 2 then rac+crprtam;

   if othotam >= 2 then rac+othotam;

   if rnbedam >= 2 then rac+rnbedam;
   if rnbtham >= 2 then rac+rnbtham; 
   if rnotham >= 2 then rac+rnotham;
   
   if roofam >= 2 then rac+roofam;
   if sidam >= 2 then rac+sidam;
   if dooram >= 2 then rac+dooram;
   if inslam >= 2 then rac+inslam;
   if pipeam >= 2 then rac+pipeam; 
   if plfxtam >= 2 then rac+plfxtam;
   if wiram >= 2 then rac+wiram;
   if securam >= 2 then rac+securam; 
   if subflam >= 2 then rac+subflam;
   
   if finflam >= 2 then rac+finflam;
   if flooram >= 2 then rac+flooram;
   if panelam >= 2 then rac+panelam;
   if cenaram >= 2 then rac+cenaram;
   if htequam >= 2 then rac+htequam;
   
   if septam >= 2 then rac+septam;
   if wtrham >= 2 then rac+wtrham;
   if dshwam >= 2 then rac+dshwam;
   if dsplam >= 2 then rac+dsplam;
   if insaam >= 2 then rac+insaam;
   if insbam >= 2 then rac+insbam;
   if inscam >= 2 then rac+inscam;
   if driveam >= 2 then rac+driveam;
   if fenceam >= 2 then rac+fenceam;
   if patioam >= 2 then rac+patioam;
   if poolam >= 2 then rac+poolam;
   if shedam >= 2 then rac+shedam;
   if outaam >= 2 then rac+outaam;
   if outbam >= 2 then rac+outbam;
   if outcam >= 2 then rac+outcam;
end;    /* end of universe check for rac */

/* degree recode */

degree = ' ';

if (coolday <= 1999) then
do;
  if (heatday > 7000) then

     degree = '1';
  else
     if (heatday >= 5500 and heatday <= 7000) then
        degree = '2';
  else
     if (heatday >= 4000 and heatday <= 5499) then
        degree = '3';
  else
     if (heatday <= 3999) then
        degree = '4';
end;
else
   if (heatday <= 1999) then
        degree = '5';
else
   if (heatday <= 3999) then
        degree = '6';
else
   put 'ILLEGAL DEGREE DAY VALUE -- control number = ' ctrlnm13;

/* metro recode - former geopuf */

metro = ' ';

if msasta80 = '1' then
   metro = '1';
else
  if msasta80 = '2' then
     if (uacode80 ge 1 and uacode80 le 9999) then
        metro = '2';
     else
       if urbrur80 = '1' then
          metro = '3';
       else
          metro = '4';
  else
       if urbrur80 = '1' then
         if (uacode80 ge 1 and uacode80 le 9999) then
            metro = '5';
         else
            metro = '6';
else
  metro = '7';

/*  recode for histry - set to 8 for adds and extras, otherwise set to a 1 */

srvyr = sysget('YRSURV');

histry = '1';
if convdate = srvyr then
   histry = '8';

/* recode for metro2 -- equals 1 when metro = 1, equals 2 when metro = 2-4, 
*                       equals 3 when metro = 5-7                        
****************  as of May 24, 1999  this recode will not be done *************
  if metro = '1' then metro2 = '1';
  if '2' <= metro <= '4' then metro2 = '2';
  if '5' <= metro <= '7' then metro2 = '3';                   */

/* recode for metro3 -- based on msasta80 and urbrur80 only  */

if msasta80 = '1' then
   metro3 = '1';
else
 if msasta80 = '2' and urbrur80 = '1' then
    metro3 = '2';
else
 if msasta80 = '2' and urbrur80 = '2' then
    metro3 = '3';
else
 if msasta80 = '3' and urbrur80 = '1' then
    metro3 = '4';
else
 if msasta80 = '3' and urbrur80 = '2' then
    metro3 = '5';

/*  finish out the top and bottom code variables: zadult, zinc, zinc2 and
    then output */

if last.ctrlnm13 then
do;
      nummor = .B;  
      /* nummor recode - number of mortgages  */
      if (status = '1' and tenure = '1') then
         do;
          if ((mg = '1') or ((mg = '2' or mg = 'B') and (hel = '1' or
             hel = 'D' or hel = 'R' or hel = ' '))) then
             do;
               nummor = .;
               hcnt = .;
               if hel = '2' then hcnt = 0;
               if hel = '1' and (1<=henum<=100) then hcnt=henum;

               if ((mg = '2' or mg = 'B') and (1<=hcnt<=100)) 
                  then nummor = hcnt;
               if ((1<=mcnt<=7) and (0<=hcnt<=100)) then
                  nummor = mcnt + hcnt;
               if nummor >=7 then nummor = 7;
               if ((hel = 'D' or hel = 'R' or hel = ' ')
                  or (henum = .D or henum = .R or henum = .)) then
                  nummor = .D;
             end;  /* end of check for mg and hel  */
         end;  /*  end of universe check for nummor */

      if zadult >= 11 then zadult = 11;
      if (zinc <= -10000 and zinc ne .B) then zinc = -10000;
      if zinc >= 999996 then zinc = 999996;
      if (zinc2 <= -10000 and zinc2 ne .B) then zinc2 = -10000;
      if zinc2 >= 999996 then zinc2 = 999996;

   output rcd.pufrcd;
   outcnt+1;
end;

if send then
do;
  put cnohh ' = number of persons obs without corresponding houshld obs';
  put cnopr ' = number of houshld obs without corresponding persons obs';
  put ctrlno ' = number of first control numbers processed';
  put ctllst ' = number of last control numbers processed';
  put incnt ' = number of view obs input';
  put outcnt ' = number of obs output';
end;

/* end of recode program */
run;


